

“麦语言” (My language) 编写

文華財經 研究部

麦语言（My language）模型开发平台

赢智的“麦语言”源于2004年文华推出的国内第一套程序化函数库，经过8年的发展，吸收几十万用户的意见反馈，一点一点完善起来的，是一套成熟稳定的模型开发平台。

麦语言倡导的是积木式的编程理念，把复杂算法封装到一个个的函数里，采用“小语法，大函数”的构建模式。语法虽然简单，但是配合专门的程序化数据结构，配合丰富的金融统计函数库，同样可以支持逻辑复杂的金融应用。

麦语言的函数库，是经常更新的，根据客户的新要求随时添加新函数，来支持编程者的交易新思想和新应用。

麦语言，是国内使用人数最多的程序化模型开发平台。

注：本课件中所用到的思路仅供参考，交易后果自负。

课程内容

- 一、模型的基本结构和跨指标模型的编写
- 二、跨周期模型的编写
- 三、模型中资金管理的编写

一、模型的基本结构和跨指标模型的编写

模型基本结构



指标和模型的区别

模型编写的语法与操作符

模型编写的结构和编写方法

```
RSV1:=CLOSE/OPEN*(20-1)/(20-1);  
CK:=SMA(RSV1,5,1);  
CD:=SMA(CK,10,1);  
CK<20&&CROSS(CK,CD),BPK;  
CK>80&&CROSS(CD,CK),SPK;
```

如何区分指标和模型


$RSV := (CLOSE - LLV (LOW, N)) / (HHV (HIGH, N) - LLV (LOW, N)) * 100;$

$K: SMA (RSV, M1, 1);$

$D: SMA (K, M2, 1);$

$J: 3 * K - 2 * D;$

指标




模型

```
RSV: = (CLOSE-LLV (LOW, N)) / (HHV (HIGH, N) -  
LLV (LOW, N)) *100;  
K: SMA (RSV, M1, 1);  
D: SMA (K, M2, 1);  
J: 3*K-2*D;
```

//以下是加入的交易指令

```
CROSS (K, D), BK; //K向上穿越D, 发出买开交易指令  
CROSS (J, 100), SP; //J向上穿越100, 发出卖平交易指令  
CROSS (D, K), SK; //K向下穿越D, 发出卖开交易指令  
CROSS (0, J), BP; //J向下穿越0, 发出买平交易指令  
AUTOFILTER;
```


交易指令



模型基本结构

指标和模型的区别



模型编写的语法与操作符

模型编写的结构和编写方法

```
RSV1:=(CLOSE-LOW)/(HIGH-LOW)*100;  
CK:=SMA(RSV1,5,1);  
CD:=SMA(CK,10,1);  
CK<20&&CROSS(CK,CD),BPK;  
CK>80&&CROSS(CD,CK),SPK;
```

MY language 编写语法:

1、命名部分:

支持汉字、字母、数字、划线格式命名，长度控制在31字符内；命名不能和已存在的公式名称重复。

2、定义变量名称

变量名称不能相互重复；

不能与参数名重复；

不能与函数名重复。

3、半角输入法的大写状态。

4、每个语句应该以分号结束。

MY language 编写语法:

5、参数部分:

可以设置六个参数;

首先是参数名称, 然后是参数的最小值, 最大值, 最后是参数的默认值;

在定义参数时要注意的是参数名称不可以重复, 12个字符内。

6、注释或者舍去

想要在编写后, 加入自己的语言注释, 在结尾处用 “//” 表示; 或者想舍去某段, 在某段在最前端加入 “//” ;

My language (麦语言) 模型开发平台 - 模型快速入门

文件 编辑 插入 设置 应用 帮助

输出 合约: IF1108 周期: 1分钟 加载

系统 自编 编辑 效果预览 收益率测算

IFELSE模型示范
REF模型示范
三减六日乖离
交叉型模型示范
交叉型模型示范
价差型模型示范
价差型模型示范
全程追踪止损示范
多条件模型示范
多条件模型示范
慢速K模型
指标转模型示范
日内高频示范1
日内高频示范2
日内高频示范3
时间函数示范
时间函数示范
模型快速入门
限价止损+巨浪止...
限价止损+限价止...

```

RSU:=(CLOSE-LLV(LOW,N))/(HHV(HIGH,N)-LLV(LOW,N))*100;
K:SMA(RSU,M1,1);
D:SMA(K,M2,1);
J:3*K-2*D;
//以下是加入的交易指令
CROSS(K,D),BK;//K向上穿越D, 发出买开交易指令
CROSS(J,100),SP;//J向上穿越100, 发出卖平交易指令
CROSS(D,K),SK;//K向下穿越D, 发出卖开交易指令
CROSS(0,J),BP;//J向下穿越0, 发出买平交易指令
0 AUTOFILTER;

```

基本信息

名称:
模型快速入门.trd 作者信息

说明:
这些示范模型,目的在于:
如何编写模型,无法不保证模
一些模型的交易结果负责。
程序化交易的基石是个性
或机构)根据自己的投资经验
后进行电脑自动交易。交易

参数列表

名称	最小	最大	默认
参数1 N	1	100	1
参数2 M1	1	100	3
参数3 M2	1	100	13
参数4	0	0	0
参数5	0	0	0
参数6	0	0	0

设置默认参数组

策略模型 模型快速入门 数字

参数

命名

MY language 操作符

操作符	意义	例
+	加法	CLOSE + OPEN 表示求收盘价及开盘价的和。
-	减法	CLOSE - OPEN 表示求收盘价及开盘价的差。
*	乘法	CLOSE * OPEN 表示求收盘价及开盘价的积。
/	除法	CLOSE / OPEN 表示求收盘价及开盘价的商。
&&	与(并且), 也可简写为AND	
	或(或者), 也可简写为OR	
>	大于	CLOSE > OPEN 表示判断当前周期是否收阳。
<	小于	CLOSE = OPEN 表示判断当前周期是否平盘。
>=	大于等于	
<=	小于等于	
<>	不等于	
=	等于操作符	
:=	只定义一个局部变量 (这个变量在画图时是不画的)	<pre>TMP1 := (OPEN + CLOSE) / 2; MA(TMP1, 10);</pre> <p>上面的公式的第一个语句定义了一个局部变量TMP1, 在下面一行中引用了这个局部变量, 但是要注意的是这个公式在画图的时候只画了第二条语句所求出的结果。</p>
:	声明了一个变量, 在画图时画出它并且按这个名字显示。	<p>相反下面这个公式则需要画出两条线, 第一条是自己定义的均价线, 同时显示了均价的名称为AVP, 第二条线是均价的简单移动平均线。</p> <pre>AVP : (OPEN + CLOSE) / 2; MA(AVP, 10);</pre>

如何运用操作符:

A: $(0+C) / 2;$

B: $C > 0;$ //判断是否收阳; 满足条件返回1, 否则返回0

D: $TIME = 0900 \&\& C > 0;$ //用于多条件逻辑关系

$MA5 = MA(C, 5);$

$MA10 = MA(C, 10);$

$CROSS(MA5, MA10);$

文华函数运用:

以参数的个数进行划分

1.求最近一次阳线到当前的周期数?

BARSLAST(C>O);

2.如果五周期均线大于十周期均线，则取五周期均线的值，否则取十周期均线的值。

IFELSE(MA5>MA10,MA5,MA10);

2、模型的基本结构

在编写前，需要将交易思想清晰量化后，通过语言函数编写完成

交易模型基本结构

1. 定义需要的每个变量
2. 交易条件(多个逻辑关系组合)+交易指令

IF指数 (CFEX 8600)日線

2766.93 37.07 / 1.36%

模型: 模型快速入门 (9, 3, 3)



模型中使用的交易指令

图示	指令	意义
↑	BK 或 BUY	买开指令
↑	BP	买平指令
↓	SK	卖开指令
↓	SP 或 SELL	卖平指令
↑	BPK	买平同时等价等量买开指令
↓	SPK	卖平同时等价等量卖开指令

IF指数 (CFEEX 8600)日线

2767.09 37.23/1.36%

模型: KDJ修改模型 (9, 3, 3)



3、跨指标模型的编写

模型中跨指标，是将多个指标交易思想结合在一起进行看盘断势。

关键词：多个交易条件

以MACD指标结合KDJ指标为例：

MACD指标结合KDJ交叉指标修正模型:

RSV: = (CLOSE-LLV (LOW, N)) / (HHV (HIGH, N) -LLV (LOW, N)) *100;

K: SMA (RSV, M1, 1);

D: SMA (K, M2, 1);

J: 3*K-2*D;

N1: =BARSLAST (CROSS (K, D)) +1;

N2: =BARSLAST (CROSS (D, K)) +1;

HH: =VALUEWHEN (N1<N2, HHV (J, N1));

LL: =VALUEWHEN (N1>N2, LLV (J, N2));

TMP: IFELSE (N1<N2, LL, HH);

CROSS (K, D) &&TMP<10, BPK;

CROSS (D, K) &&TMP>90, SPK;

AUTOFILTER;

KDJ修正模型

——》模型中加入MACD指标思路:

IF指数 (CFEX 8600) 日线 2778.59 48.73 / 1.79%

模型: KDJ MACD (9, 3, 3, 12, 26, 9)



二、跨周期模型的编写

跨周期函数介绍

引用某品种在某个周期上加载了某个指标的数据。

用法：

```
#IMPORT [CODE, PERIOD, FORMULA] AS VAR
```

引用 **CODE** 所对应的合约 **PERIOD** 周期下指标 **FORMULA** 的数据。

CODE 文华码，**PERIOD** 周期，**FORMULA** 引用指标名，
VAR 定义变量名

跨周期跨合约模型的编写规则

- 1.只能引用 .FML/.XFML文件
- 2.只能引用如下周期：MIN1 MIN3 MIN5 MIN15 MIN30
HOUR1 DAY WEEK MONTH
- 3.只能短周期引用长周期
- 4.被引用的指标中不能存在引用
- 5.如果不写文华码，默认引用当前合约,也可以直接写合约代码如:rb1201
- 6.FORMULA 引用指标名，只能引用除数字、或者数字开头的名称之外的指标名称。

例 同一合约不同周期的数据调用

要求

- 当30分钟均线出现多头排列时，5分钟KD线金叉，做多
- 当30分钟均线出现多头排列时，5分钟KD线死叉，平多
- 当30分钟均线出现空头排列时，5分钟KD线死叉，做空
- 当30分钟均线出现空头排列时，5分钟KD线金叉，平空

先建立一个指标 名称AAA

```
MA5:=MA(C,5);
```

```
MA10:=MA(C,10);
```

```
MA30:=MA(C,30);
```

再建立模型加载在5分钟周期

```
#IMPORT [ , MIN30,AAA] AS VAR
```

```
DM5:=VAR.MA5;
```

```
DM10:=VAR.MA10;
```

```
DM30:=VAR.MA30;
```

```
RSV:=(CLOSE-LLV(LOW,N))/(HHV(HIGH,N)-LLV(LOW,N))*100;
```

```
K:SMA(RSV,M1,1);
```

```
D:SMA(K,M2,1);
```

```
J:3*K-2*D;
```

```
DM5>DM10&&DM10>DM30&&CROSS(K,D),BK;
```

```
DM5>DM10&&DM10<DM30&&CROSS(D,K),SP;
```

```
DM5<DM10&&DM10<DM30&&CROSS(D,K),SK;
```

```
DM5<DM10&&DM10>DM30&&CROSS(K,D),BP;
```

```
AUTOFILTER;
```

例 同一合约不同周期的数据调用

要求

- 30分钟周期上，当前面一根MA5大于MA10，并且5分钟周期上，MA5上穿MA10，做多。
- 30分钟周期上，当前面一根MA5大于MA10，并且5分钟周期上，MA5下穿MA10，做空。
- 尾盘平仓

关键：不同周期共振，日内交易

先建立一个指标 名称AAA

```
RMA5:=REF(MA(C,5),1);
```

```
RMA10:=REF(MA(C,10),1);
```

再建立模型加载5分钟周期

```
#IMPORT[,MIN30,AAA] AS VAR
```

```
N:BARSLAST(DATE<>REF(DATE,1))+1;
```

```
DM5:=VAR.RMA5;
```

```
DM10:=VAR.RMA10;
```

```
MA5:=MA(C,5);
```

```
MA10:=MA(C,10);
```

```
DM5>DM10&&((CROSS(MA5,MA10)&&N>1&&TIME<1450)||((MA5>MA10&&N=1  
)),BPK;
```

```
TIME>=1450,SP;
```

```
DM5<DM10&&((CROSS(MA10,MA5)&&N>1&&TIME<1450)||((MA5<MA10&&N=1  
)),SPK;
```

```
TIME>=1450,BP;
```

```
AUTOFILTER;
```

思考：
不同合约的数据如何调用

三、模型中资金管理的编写

1、头寸函数函数介绍

ISLASTBK	判断上一个交易信号是否是BK。 用法：ISLASTBK 如果上一个交易信号是BK则返回1（Yes），否则返回0（No）
ISLASTSK	判断上一个交易信号是否是SK。 用法：ISLASTSK 如果上一个交易信号是SK则返回1（Yes），否则返回0（No）
ISLASTBP	判断上一个交易信号是否是BP。 用法：ISLASTBP 如果上一个交易信号是BP则返回1（Yes），否则返回0（No）
ISLASTSP	判断上一个交易信号是否是SP。 用法：ISLASTSP 如果上一个交易信号是SP则返回1（Yes），否则返回0（No）
ISLASTBPK	判断上一个交易信号是否是BPK。 用法：ISLASTBPK 如果上一个交易信号是BPK则返回1（Yes），否则返回0（No）
ISLASTSPK	判断上一个交易信号是否是SPK。 用法：ISLASTSPK 如果上一个交易信号是SPK则返回1（Yes），否则返回0（No）

<p>BARSBK</p>	<p>上一次买开信号位置</p> <p>用法:</p> <p>BARSBK返回上一次买开仓距离当前k线的k线数。</p>
<p>BARSSK</p>	<p>上一次卖开信号位置</p> <p>用法:</p> <p>BARSSK返回上一次卖开仓距离当前k线的k线数。</p>
<p>BARSBP</p>	<p>上一次买平信号位置</p> <p>用法:</p> <p>BARSBP返回上一次买平仓距离当前k线的k线数。</p>
<p>BARSSP</p>	<p>上一次卖平信号位置</p> <p>用法:</p> <p>BARSSP返回上一次卖平仓距离当前k线的k线数。</p>

买开信号位置的买开信号价位。

用法: **BKPRICE**返回最近一次模型买开位置的买开信号价位。

例如: **BKPRICE-CLOSE>60 && BKPRICE>0, SP;**//如果买开价位比当前价位高出60,且买开价位存在, 卖平仓

BKPRICE 请注意当模型存在连续多个开仓信号(加仓)的情况下, 该函数返回的是最后一次开仓信号的价格,而不是开仓均价。

注: 过滤模型的**BKPRICE** 只在加载之后的K线上才返回信号价位, 历史K线信号由于无信号价位会返回0, 使用时请注意判断**BKPRICE>0**。非过滤模型加载时的历史K线上、效果测试中该函数返回信号位置的收盘价

卖开信号位置的卖开信号价位

用法: **SKPRICE**返回最近一次模型卖开位置的卖开信号价位。

例如:**CLOSE-SKPRICE>60 && SKPRICE>0, BP;**//如果当前价位高出卖开价位60, 且卖开价位存在, 买平仓

SKPRICE 请注意当模型存在连续多个开仓信号(加仓)的情况下, 该函数返回的是最后一次开仓信号的价格,而不是开仓均价。

注: 过滤模型的**SKPRICE** 只在加载之后的K线上才返回信号价位, 历史K线信号由于无信号价位会返回0, 使用时请注意判断**SKPRICE>0**。非过滤模型加载时历史K线上、效果测试中该函数返回信号位置的收盘价

FEE	<p>合约手续费</p> <p>用法: FEE返回当前合约的手续费(用户启动模组时设置的)。</p> <p>注意不能与未来函数同时使用如 ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX 等</p> <p>本函数运算量很大,将占用很多的CPU资源,导致行情刷新速度变慢,请谨慎使用!</p>
MONEY	<p>虚拟资金余额</p> <p>用法: MONEY返回虚拟资金余额。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>
MARGIN	<p>合约保证金</p> <p>用法: MARGIN返回当前合约的保证金比率(用户启动模组时设置的)。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>
VOLMARGIN	<p>模型虚拟持仓保证金</p> <p>用法: VOLMARGIN计算当前的持仓保证金。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>

<p>MONEYRATIO</p>	<p>资金使用率</p> <p>用法: MONEYRATIO返回当前的虚拟资金的使用率。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>
<p>MONEYTOT</p>	<p>虚拟总资金</p> <p>用法: MONEYTOT返回当前虚拟总资金（虚拟资金余额+持仓保证金）。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>
<p>PROFIT</p>	<p>虚拟逐笔浮盈</p> <p>用法: PROFIT返回当前的虚拟逐笔浮动盈亏。</p> <p>注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS,TROUGH,TROUGHBARS,REFX等可能会导致误差。</p>
<p>SETDEALPERCENT</p>	<p>设置下单的虚拟资金使用比例</p> <p>用法: SETDEALPERCENT(fPercent)表示每次按资金的fPercent（范围1~100）下单。</p> <p>例子: SETDEALPERCENT(20); //每次按资金比例的%20下单</p> <p>注: 应该与AUTOFILTER函数同时使用</p>

模型虚拟多头持仓

用法:

BUYVOL BUYVOL返回模型虚拟多头持仓。

注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS, TROUGH,TROUGHBARS,REFX等可能会导致误差。

模型虚拟空头持仓

用法:

SELLVOL SELLVOL返回模型虚拟空头持仓。

注意与未来函数同时使用ISLASTBAR,EMA2,ZIGZAG,BACKSET,PEAK,PEAKBARS, TROUGH,TROUGHBARS,REFX等可能会导致误差。

2、利用头寸函数实现对仓位的加减。

注意，交易时要考虑前一信号方向防止锁仓。

例：资金管理模型-加减仓模型

A:=多头交易条件;

B:=空头交易条件;

E:=多头平仓条件;

F:=空头平仓条件;

A && NOT(ISLASTBK)&& NOT(ISLASTSK),BK(2);

B && NOT(ISLASTBK)&& NOT(ISLASTSK),SK(2);

BUYVOL>2 && A && BARSBK>1,BK(1);

SELLVOL>2 && B && BARSSK>1,SK(1);

E && ISLASTBK,SP(BUYVOL);

F && ISLASTSK,BP(SELLVOL);

注意：信号消失不能全部平仓造成锁仓现象。

3、利用**BKPRICE**和 **SKPRICE**等函数。编写限价止盈、限价止损和追踪止损。

例：限价止损、限价止盈模型

A:=多头交易条件;

B:=空头交易条件;

E:=多头平仓条件;

F:=空头平仓条件;

A,BK;

E||C<=BKPRICE-100||C>=BKPRICE+150,SP;

B,SK;

F||C>=SKPRICE+100||C<=SKPRICE-150,BP;

AUTOFILTER;

例：追踪止损

JW:=5; //定义最小价位

ZS:=10; //定义止损为10个最小价位

BC:=5; //定义步长为5个最小价位

MA1 := MA(C,5); //5周期均线

H1:=HHV(H,BARSBK+1); //取自开仓K线到现在的最高价

C>MA1,BK; //收盘价大于5周期均线，开仓

C<MA1||(C<=(BKPRICE-(ZS*JW)))+(BC*JW)*FLOOR((H1-BKPRICE)/(BC*JW)), SP;

//收盘价小于5周期均线，或者满足追踪止损条件，平仓

例：追踪止损

```
JW:=MINPRICE('RU1201'); //定义最小变动价位
ZS:=10; //定义止损为10个最小价位
BC:=5; //定义步长为5个最小价位
MA1 := MA(C,5); //5周期均线
H1:=HHV(H,BARSBK+1); //取自开仓K线到现在的最高价
L1:=LLV(L,BARSSK+1); //取自开仓K线到现在的最低价
BS:=IFELSE(H1>BKPRICE,(BKPRICE-(ZS*JW))+(BC*JW)*FLOOR((H1-
    BKPRICE)/(BC*JW)),(BKPRICE-(ZS*JW))); //多头止损价位
SS:=IFELSE(L1<SKPRICE,(SKPRICE+(ZS*JW))-(BC*JW)*FLOOR((SKPRICE-
    L1)/(BC*JW)),(SKPRICE+(ZS*JW))); //空头止损价位
C>MA1,BPK; //收盘价大于5周期均线，开仓
C<=BS,SP; //满足追踪止损条件，平仓
MA1>C,SPK;
C>=SS,BP; //满足追踪止损条件，平仓
AUTOFILTER;
```

祝交易顺利

谢 谢